# $Id: PKGBUILD 67786 2012-03-14 12:26:21Z giovanni $
# Maintainer: Jaroslav Lichtblau <dragonlord@aur.archlinux.org>
# Contributor: Jeff Mickey <jeff@archlinux.org>

pkgname=surfraw
pkgver=2.2.8
pkgrel=2
pkgdesc="Shell Users' Revolutionary Front Rage Against the Web"
arch=('any')
url="http://surfraw.alioth.debian.org/"
license=('custom')
depends=('perl')
install=$pkgname.install
backup=('etc/xdg/surfraw/conf' 'etc/xdg/surfraw/bookmarks')
source=(http://surfraw.alioth.debian.org/dist/$pkgname-$pkgver.tar.gz)
md5sums=('e0f571f7a2d109555c26bdb40781a3f2')

build() {
  cd ${srcdir}/$pkgname-$pkgver

  ./configure --prefix=/usr --sysconfdir=/etc --disable-opensearch
  make 
}

package () {
  cd ${srcdir}/$pkgname-$pkgver

  make DESTDIR=${pkgdir} install

  # license
  install -D -m644 COPYING ${pkgdir}/usr/share/licenses/$pkgname/COPYING 
}
